An Application of Rational Trees in a Logic Programming Interpreter for a Procedural Language

Manuel Carro
C.S. Dept., Technical University of Madrid
mcarro@fi.upm.es
February 1, 2008

Abstract
We describe here a simple application of rational trees to the implementation of an interpreter for a procedural language written in a logic programming language. This is possible in languages designed to support rational trees (such as Prolog II and its descendants), but also in traditional Prolog, whose data structures are initially based on Herbrand terms, but in which implementations often omit the occurs check needed to avoid the creation of infinite data structures.

We provide code implementing two interpreters, one of which needs non-occurs-check unification, which makes it faster (and more economic). We provide experimental data supporting this, and we argue that rational trees are interesting enough as to receive thorough support inside the language.

1 Introduction

Occurs check, i.e., the need to verify whether a variable \( X \) occurs into a term \( T \) prior to performing the unification \( X = T \) (and failing if that is the case), is needed in order to provide soundness to logic programs based on first order logic. The need to check this can appears in general, when two terms have to be unified, as equations of the above form can be generated while unifying them. For example, solving

\[
f(X, X) = f(g(Y), Y)
\]

can be reduced to

\[
\begin{align*}
X &= g(Y) \\
X &= Y
\end{align*}
\]

which in turn boils down to \( X = g(X) \). The relevance of occurs check when reasoning on programs supposedly executing over the Herbrand domain can be seen from the following examples, taken from [CKS96]:

Correspondence with First-Order Semantics
Consider the statement \( \exists z : p(z, z) \). It is not a logical consequence of \( \forall x \exists y : p(x, y) \) (this is easy to see by taking \( p(x, y) \) as, e.g., \( y = x + 1 \) in the domain of the natural numbers). However, in the Prolog program
q :- p(X, X).
\[ p(X, Y) :\neg Y = s(X). \]

the query \(- q\). will succeed if occurs-check is not performed (as it is the default case\(^1\) with many current Prolog compilers and interpreters). This means that Prolog without occurs-check is not adequate as a theorem prover for first-order logic.

**Program Transformation** Some simple program transformations rely on occurs-check being performed, or else the transformed program can behave differently from the original one. For example, the clause
\[ q(X, Y, U, V) :\neg X = Y, X = a, U = f(X), V = f(Y). \]

can be trivially transformed into
\[ q'(X, Y, f(X), f(Y)) :\neg X = Y, X = a. \]

The call
\[- q(X, Y, X, Y). \]

finishes with failure in all Prolog systems, but the call
\[- q'(X, Y, X, Y). \]

does not terminate in some Prolog systems, depending on how unification of circular terms is handled.\(^2\)

Besides ensuring soundness, the assumption of the existence of occurs check makes it possible to carry around unification equations in solved form, i.e., in the form of a set of equations of the form \( Var_i = Term_i \) where no \( Var_i \) occurs inside any \( Term_j \). This permits, for example, projecting equations over variables (i.e., what a toplevel does to print answers) in a relatively straightforward way. Occurs-check is however not without a cost: checking for occurrences is, using a naive algorithm, quadratic with respect to the size of the terms to be unified. More refined algorithms can reduce the complexity to \( O(n) \) \( [\text{MM76, MM82, PW78}] \), but requiring large memory space or with a linear constant too high. This has somehow prevented the widespread use of occurs-check unification. Quoting \( [\text{ALP93}] \):

An extreme example is \texttt{append/3} for difference lists. Without the occurs check, the cost is \( O(1) \); with the occurs check, the cost is \( O(n) \) on the size of the terms (in other words, it gets the same cost as regular \texttt{append/3}). Normally, the programmer knows\footnote{It is to be noted that the ISO Prolog standard \( [\text{DEDC96}] \) does require the presence of a builtin \texttt{(unify_with_occurs_check/2)} able to perform unification with occurs check.} that the occurs check is not needed, so this is a big win. \footnote{In particular, we have verified that Yap-5.4.1 and SWI-5.2.7 did not terminate, but SICStus and Ciao Prolog (which share an original common code base), did. Of course, newer versions of Yap and SWI may have already filled in this gap.}
Benchmarking of Prolog systems which have included support for occurs check reports modest overheads in the execution (around 10%-15% of global speed loss, with some studies reporting only 5% degradation [DRSW89]). While this might be seen as a relatively small price to pay, one might wonder if is worth paying it every time a unification takes place, especially taking into account that the programmer often knows where cyclic terms can be generated. Moreover, cyclic terms bring several useful characteristics, both in expressiveness and in performance, and we argue that they should be properly supported. Note that leaving the decision of performing or not occurs check to some analysis is not satisfactory, as the programmer may require it or not, so some sort of annotation guiding the compiler will be needed in that case. The analyzer might otherwise decide to include occurs check unification where they are not intended. In fact, as we will see, deciding where these are necessary is not straightforward, and different applications can be at odds with respect to the required behavior of, e.g., some library predicates.

Not using occurs check at all changes the term universe, which must therefore be given a proper semantics with respect to unification and, if needed, with respect to difference constraints. This has been studied elsewhere [Col84, Col82a, JS86], and the resulting term universe has been named rational trees, as opposed to Herbrand terms. Rational trees are part of the design of, for example, the Marseilles family of Prolog systems [Col82b, Col90, PRO, Col96, BT95]. Even if infinite, in the sense that there are paths of infinite length, rational trees can be finitely represented and used to represent cyclic data structures.

There are practical reasons to allow the creation of cyclic terms. In practice, it seems that using rational trees does not need more resources than using Herbrand terms, and the necessary algorithms (not only unification, but also garbage collection routines, term-copying, and database modification predicates) can be effectively implemented. Rational trees have been explicitly used in graphics [EC83], parser generation and grammar manipulation [Col82a, GC84], computing with finite-state automata [Col82a], and several areas of natural language processing: rational trees are used in implementations of the HPSG (Head-driven Phrase Structure Grammar) formalism [PS84], in the ALE (Attribute Logic Engine) system [Car92], and in the ProFIT (Prolog with Features, Inheritance and Templates) system [Erb95].

In this paper we present yet another application of rational trees: the implementation of interpreters for procedural languages featuring explicit control instructions (such as jumps, if-then-elsees, whilees and repeates) which can divert the execution flow. Our aim is to implement an interpreter which is simple and easy to verify, but without sacrificing performance unnecessarily. Also, with the aim of producing an interpreter amenable to be analyzed and transformed, we want it to be as clean and trick-free possible. Some of the techniques we present can be performed using attribute variables [Hol92] and occurs-check unification, but the resulting program is less amenable to automatic analysis and transformation. Describing it falls beyond the scope of this paper.

In the rest of the paper we will first describe our initial problem (Section 2) and then we will describe and evaluate two solutions (Sections 3 and 4). We include code for the interpreters developed and for the benchmark programs.
Table 1: Syntax and (short) semantics of a simple language

2 The Problem and the Target Language

We want to interpret a procedural language using a (high-level) logic language (Prolog, in our case); there are several reasons to perform such interpretation:

- We may want to execute programs in a language for which there is no compiler or interpreter available, and we do not want to invest time in writing a more sophisticated compiler. A simple interpreter might be enough for our purposes.

- We may want to experiment with alternative semantics for the same (syntactically speaking) language. It is clearly much easier to do that if the interpreter is easy to understand and change.

- We may want to apply analyzers written for the interpreter language to the interpreted language. This is possible by first unfolding the interpreter with respect to the interpreted program, analyzing the resulting unfolded program, and then reinterpreting the analysis in the domain of the interpreted program.

We will first set down a toy target language which notwithstanding features all the characteristics we need. For simplicity we will adopt an assembler-like language whose syntax and semantics are easy to describe. A summary of the (minimal) set of instructions we have chosen is shown in Table 1. Figures 9, 10, and 11 show several programs in the target language (those we have used to evaluate our implementations).

The target machine has a single register, the accumulator, which is used in all but two instructions, and which acts as implicit source and destination for data. Explicit data sources are either numeric constants (which appear directly in the code) or symbolic labels, which represent locations where the operands are stored. Data labels do not need to be declared, as
the assembler / interpreter assigns them a location. Data is loaded from implicit operands or from addresses into the accumulator; data in the accumulator can be stored in memory addresses represented by labels. Arithmetic operations involve the accumulator both as source of values and as destination of results. The second operand for arithmetic instructions is explicit in the program source.

Instructions can be labeled with a symbolic name which is used by control instructions; an unbound number of such label names are available. These labels appear as operands of jumps; unconditional jumps make the execution continue at the code associated to the label. Conditional jumps test the accumulator value and decide whether to continue executing on the next instruction of the program, or if execution has to divert to the instruction whose label is the operand of the jump instruction. There is no state register (usual in real microprocessors) which changes with the accumulator operations. There is also a no-op instruction. Execution starts in the first instruction of the program, and finishes when attempting to continue executing after the last instruction. There is no halt instruction.\(^3\)

\begin{verbatim}
program(fibo,  
  [jnez(calculate), load(0), sto(curr), jmp(end), calculate:sto(ind),  
    load(0), sto(prev), load(1), sto(curr), start_loop:load(ind), sub(1),  
    sto(ind), jez(end), load(curr), sto(inter), add(prev), sto(curr),  
    load(inter), sto(prev), jmp(start_loop), end:load(curr)]).
\end{verbatim}

Figure 1: Prolog representation of program in Figure 10

We will assume that programs are already in the form of Prolog terms. Figure 1 represents the Fibonacci program in Figure 10 as a fact, and every instruction is assigned to a Prolog functor. The interpreter and programs are designed to be used as

\begin{verbatim}
?- program(fibo, Program), interpret(Program, 400, A).
\end{verbatim}

where 400 is the initial value of the accumulator and A is the contents of the accumulator at the end of the execution. By convention we take the contents of the accumulator at the beginning of the execution as the input for the interpreted program, and its value at the end as the output value of the program, and we expect the programs to abide by this convention. We also assume that the programs are correct: for example, all labels to jump to appear in the program, there are no repeated labels, etc. The interpreter will otherwise fail.

3 An Interpreter in the Herbrand Domain

The two tasks the interpreter has to perform when executing the instructions are maintaining the memory state (including the accumulator) and deciding the next instruction to execute (and doing so). Conceptually, the memory is a mapping

\begin{verbatim}
label \mapsto value
\end{verbatim}

which relates every memory address name to a single value. This mapping is updated with every program instruction which stores a value into memory. Its implementation can be left, as we will do, to a table which takes the form of a Prolog dictionary, with the operations in table 2.

\(^3\)It may be replaced by making the last program instruction a nop and jumping to it.
interpret_nv(Program, AcIn, AcOut):-
    dic_empty(DicIn),
    interpret_naive(Program, Program, DicIn, _DicOut, AcIn, AcOut).

interpret_naive([], _Program, _, _, Acum, Acum).
interpret_naive([LI|Is], Program, DictIn, DictOut, AIn, AOut):-
    remove_label(LI, I),
    execute(I, Is, Program, NProgram, DictIn, DictMid, AIn, AMid),
    interpret_naive(NProgram, Program, DictMid, DictOut, AMid, AOut).

execute(load(NumOrLabel), Is, _Program, Is, D, D, AcIn, AcOut):-
    eval(NumOrLabel, D, Value).
execute(add(NumOrLabel), Is, _Program, Is, D, D, AcIn, AcOut):-
    eval(NumOrLabel, D, Value),
    AcOut is AcIn + Value.
execute(sub(NumOrLabel), Is, _Program, Is, D, D, AcIn, AcOut):-
    eval(NumOrLabel, D, Value),
    AcOut is AcIn - Value.
execute(sto(Label), Is, _Program, Is, DIn, DOut, Acum, Acum):-
    dic_replace(DIn, Label, Acum, DOut).
execute(jmp(Label), _, Program, Is, D, D, A, A):-
    Is = [Label:_[_] | _], append(_, Is, Program).
execute(jez(Label), NIs, Program, Is, D, D, A, A):-
    (A = 0 -> Is = [Label:_[_] | _], append(_, Is, Program) ; Is = NIs).
execute(jnez(Label), NIs, Program, Is, D, D, A, A):-
    (A \== 0 -> Is = [Label:_[_] | _], append(_, Is, Program) ; Is = NIs).
execute(nop, Is, _Program, Is, D, D, A, A).

remove_label(_:I, I):-. eval(Number, _Dict, Number):-
    number(Number).
remove_label(I, I).
    eval(Label, Dict, Number):-
    atom(Label),
    dic_get(Dict, Label, Number).

Figure 2: An interpreter which does not need rational trees

There is a similar relationship between program labels and program code: every program label maps into the sequence of instructions starting at the labeled one. With the current language, this mapping does not change and is explicit in the program text, so it is possible to search in the program text for the label to jump to when needed. Figure 2 portrays the complete code of an interpreter for the target language, which assumes that programs are already represented as in Figure 1.

interpret_nv/2 traverses the program executing its instructions as they are found. When an instruction which has to store data in memory is found, the table which implements the memory is updated. The accumulator is explicitly passed around in a pair of arguments, corresponding to its state before and after executing each instruction. It could have been assigned a (designated) location in the table which represents the memory; however, as the accumulator is used in almost all instructions, having a direct access to it is reasonable and easy to code. Jumping to a label is performed by searching the corresponding point in the program. This makes the jump operation to have complexity $O(n)$, where $n$ is the size (in instructions) of the program. This overhead does not affect the run-time complexity of the interpreted program, since the overhead is constant for every program, and does not change
Table 2: Table operations

with the input arguments.

We will not prove formally that this interpreter does not generate rational trees, but we will give some intuition supporting this. In the intended use of interpret_nv/3, the Program input argument is ground and AcIn should also be ground, and both should be Herbrand terms. Therefore, in the interpret_naive/6 predicate the first, second, and fourth predicates should be ground. All the operations take place in execute/8, and in this predicate arguments 1 to 4 and 7 are ground at call time. We need of course the code of the dictionary predicates to make sure that no cyclic terms are generated inside it, but we can give this for granted with a suitable implementation. The dictionary is used only to store (ground) values associated to (ground) keys and to retrieve them. Thus, the only unifications which take place in the interpreter have a ground term on one side of the equation, and no cyclic data structures can be generated.

On the other hand it is straightforward to make this program generate rational trees with a non-intended query: simply calling it as

?- interpret_nv([[sto(where)], X, f(X)].

will create a cyclic term.

Table 3 shows the execution times for the interpretation of the programs in figures 9, 10, and 11. The same data is depicted in Figure 3 (where the size of the input data has been scaled to the range [1.5, 7.0] so that the three graphs fit into the same figure). Time is in milliseconds, and was measured as the average of several runs. That figure alone is not really interesting, as it only shows the relative speed of different programs which makes different calculations.

|               | square |               | Fibonacci |               | factorial |
|---------------|--------|---------------|-----------|---------------|-----------|
| Input size    | Time   | Input size    | Time      | Input size    | Time      |
| 40000         | 487    | 20000         | 277       | 300           | 484       |
| 45000         | 532    | 25000         | 360       | 350           | 670       |
| 50000         | 590    | 30000         | 449       | 400           | 880       |
| 55000         | 650    | 35000         | 532       | 450           | 1107      |
| 60000         | 707    |               |           | 500           | 1377      |
| 65000         | 774    |               |           | 550           | 1682      |

Table 3: Execution times, interpreter free of occurs-check
The overhead brought about by the search of the code to jump to can be diminished in several ways. Performance can be improved by, for example, storing program fragments in a table where labels are keys, and jumps would be preceded by a search for the correct fragment of code:

\[
\text{execute}(\text{jmp}(\text{Label}), _, \text{Code}, D, D, A, A):- \\
\quad \text{dic_get}(D, \text{Label}, \text{Code}).
\]

Code has had to be associated to Label in a previous traversal of the program. However, this code performs an unnecessary search, since the representation of jump instructions (in fact, of every instruction) can be augmented to contain the code to be executed next, very similarly to a continuation passing programming style. Conditional jump instructions need two addresses for the then and the else branches, and the rest of the instructions need just one. For example, the program fragment

\[
[\text{sto}(\text{ind}), \text{load}(0), \text{sto}(\text{prev}), \text{load}(1), \text{sto}(\text{curr}), \text{load}(\text{ind})]
\]

which does no have any jump instruction is translated by the predicate thread_program/2 (Figure 5) into

\[
\text{sto}(\text{ind}, \text{load}(0, \text{sto}(\text{prev}, \text{load}(1, \text{sto}(\text{curr}, \text{load}(\text{ind}, \text{end})))))
\]
interpret_thr(Program, AcIn, AcOut):-
dic_empty(ProgramDicIn),
thread_program(Program, ProgramDicIn, Threaded),
dic_empty(MemDicIn),
interpret_threaded(Threaded, MemDicIn, _MemDicOut, AcIn, AcOut).

thread_program([], _, end).
thread_program([Inst|Next], Dict, Threaded):-
thread_instruction(Inst, Dict, Threaded, RestThreaded),
thread_program(Next, Dict, RestThreaded).

thread_instruction(NewLabel:Instruction, Dict, Threaded, RestThread):-
dic_lookup(Dict, label:Label, Threaded),
thread_instruction(Instruction, Dict, Threaded, RestThread).

thread_instruction(load(What), _, load(What, Next), Next).
thread_instruction(sto(What), _, sto(What, Next), Next).
thread_instruction(add(What), _, add(What, Next), Next).
thread_instruction(sub(What), _, sub(What, Next), Next).
thread_instruction(nop, _, nop(Next), Next).
thread_instruction(jmp(Label), Dict, jmp(Code), _Next):-
dic_lookup(Dict, label:Label, Code).
thread_instruction(jez(Label), Dict, jez(CodeYes, CodeNo), CodeNo):-
dic_lookup(Dict, label:Label, CodeYes).
thread_instruction(jnez(Label), Dict, jnez(CodeYes, CodeNo), CodeNo):-
dic_lookup(Dict, label:Label, CodeYes).

Figure 5: Threading the code into a rational tree

where the list structure is not needed any longer, since continuations are explicit in the representation of the instructions, and the pseudo-instruction end takes the place of the [] ending the program. Program fragments which have forward conditional jumps generate trees: the code

[jez(is_zero), load(in), is_zero:sub(acum), sto(out)]

translates into the tree (Figure 7)

jez(sub(acum,sto(out,end)),load(in,sub(acum,sto(out,end))))

where the subtree sub(acum,sto(out,end)) appears twice and the labels are not needed any longer.4

A code which performs a jump backward in the code, such as

[loop:load(data),sub(1),sto(data),jnez(loop),load(in)]

is translated into the rational tree (Figure 8)

load(data,sub(1,sto(data),jnez(load(data,sub(1,sto(...))),load(in,end))))

where the dots abstract an infinite tree, but the whole term is a rational tree. A closer look at how threading is done reveals that the unification which constructs the cyclic term is performed inside the table implementation, in the first pass of the process which builds an (intermediate) table associating labels with code. A trivial program such as loop:jmp(loop)

4In any modern Prolog system these two subtrees will be internally represented physically as the same data structure, so there is no waste of memory.
interpret_threaded(end, _, _, Acum, Acum).
interpret_threaded(Instr, DictIn, DictOut, AIn, AOut):-
    execute_threaded(Instr, Cont, DictIn, DictMid, AIn, AMid),
    interpret_threaded(Cont, DictMid, DictOut, AMid, AOut).
execute_threaded(load(NumOrLabel, Cont), Cont, D, D, _, Value):-
    eval(NumOrLabel, D, Value).
execute_threaded(add(NumOrLabel, Cont), Cont, D, D, AcIn, AcOut):-
    eval(NumOrLabel, D, Value), AcOut is AcIn + Value.
execute_threaded(sub(NumOrLabel, Cont), Cont, D, D, AcIn, AcOut):-
    eval(NumOrLabel, D, Value), AcOut is AcIn - Value.
execute_threaded(sto(Label, Cont), Cont, DIn, DOut, Acum, Acum):-
    dic_replace(DIn, Label, Acum, DOut).
execute_threaded(jmp(Cont), Cont, D, D, A, A).
execute_threaded(jez(Yes, No), Cont, D, D, A, A):-
    (A = 0 -> Cont = Yes ; Cont = No).
execute_threaded(jnez(Yes, No), Cont, D, D, A, A):-
    (A ~= 0 -> Cont = Yes ; Cont = No).
execute_threaded(nop(Cont), Cont, D, D, A, A).

Figure 6: Interpreter for a rewritten program

\[
\begin{aligned}
\text{jez(} & \text{•, load(in,} & \text{•}) \\
\text{sub(acum,sto(out,end))} & \text{sub(acum,sto(out,end))}
\end{aligned}
\]

Figure 7: Tree structure for a threaded program with jumps forward

generates a rational tree due to the following table accesses and unifications (distilled from the execution of the interpreter):

dic_lookup(Dict, loop, Threaded),
Threaded = jmp(Code),
dic_lookup(Dict, loop, Code)

which actually executes \( T = \text{jmp(C)} \), \( T = C \) inside the code of \text{dic_lookup/3}.

In return for the first pass which threads the code constructing the rational tree, the interpreter in Figure 6 is simpler than the one in Figure 2 as no searches for the labels are necessary: the code is directly accessible (in \( O(1) \)) in every instruction. This should bring performance advantages; this assumption is supported by the numbers put forward in Table 4 and shown in Figure 6 which are consistently better than the ones for the previous interpreter.

\[
\text{load(data, sub(1, sto(data, jnez(\ldots, load(in, end))))})
\]

Figure 8: Rational tree for a jump backwards
5 Conclusions

We have shown two interpreters for an assembler language which differ in the way jump instructions are handled. The former performs linear searches (which can however be improved), and the latter performs a first pass which transforms the code into a rational tree which makes it possible to execute jumps in $O(1)$. We have evaluated the performance of both interpreters, and the interpreter using rational trees outperforms considerably the Herbrand-based one.

Therefore, using rational trees can bring benefits in terms of performance, and supporting them correctly in Prolog systems should make their use (and that of associated programming techniques) more widespread. However, giving full support is not straightforward, as many builtins will have to be ready to accept and process them correctly. The programming style is also affected, because base cases in recursion loops are not guaranteed to be reached, as they were with (ground) Herbrand terms. Additionally, there is no a priori way to detect, from the level of first order logic, that the predicate is looping with current LP systems.

On the other hand, it does not seem to be a good idea to protect indiscriminately library predicates against creation of cyclic terms, for these predicates might be used to construct them on purpose, as in our case. It is true that hidden occurs checks can in some cases be overcome if desired, by extracting culprit unifications, e.g., by making

\[
dic\_lookup(Dict, \text{loop}, \text{Threaded}),
Threaded = \text{jmp}(Code),
dic\_lookup(Dict, \text{loop}, \text{ThreadedCode}),
Code = \text{ThreadedCode}
\]

instead of the code shown in section 4. This kind of transformations, which are correct if the called predicates are pure, can however lead to termination (and correctness) problems in non-pure predicates, and they have to be taken with a grain of salt. We deem much more useful to let the programmer to switch occurs-check on or off if necessary, maybe aided or guided by the results of program analysis.

References

[ALP93] The Occurs Check. ALP Newsletter, Volume 6/2, May 1993. Available at http://www-lp.doc.ic.ac.uk/UserPages/staff/ft/alp/net/impl/occur.html.

[BT95] Frédéric Benhamou and Touraïvane. Prolog IV: langage et algorithmes. In JFPL'95: IVèmes Journées Francophones de Programmation en Logique, pages 51–65, Dijon, France, 1995. Teknea.

[Car92] B. Carpenter. Cambridge Tracts in Theoretical Computer Science, volume 32, chapter The Logic of Typed Feature Structures with Applications to Unification-Based Grammars, Logic Programming and Constraint Resolution. Cambridge University Press, 1992.

[CKS96] Lobel Crnogorac, Andrew D. Kelly, and Harald Sondergaard. A comparison of three occur-check analysers. In Static Analysis Symposium, pages 159–173, 1996.
[Col82a] A. Colmerauer. Prolog and infinite trees. In K.L. Clark and S.-A. Tärnlund, editors, *Logic Programming*, pages 231–251. Academic Press, 1982.

[Col82b] A. Colmerauer et al. *Prolog II: Reference Manual and Theoretical Model*. Groupe D’intelligence Artificielle, Faculté Des Sciences De Luminy, Marseille, 1982.

[Col84] Alain Colmerauer. Equations and inequations on finite and infinite trees. In *FGCS-84: Proceedings International Conference on Fifth Generation Computer Systems*, pages 85–99, Tokyo, 1984. ICOT.

[Col90] Alain Colmerauer. An Introduction to Prolog III. *Communications of the ACM*, 33(7):69–90, July 1990.

[Col96] Alain Colmerauer. Specifications of Prolog IV. Draft, 1996.

[DEDC96] P. Deransart, A. Ed-Dbali, and L. Cervoni. *Prolog: The Standard*. Springer-Verlag, 1996.

[DRSW89] P. Dreussen, W. Rosentiel, K. E. Schauser, and J. Wedeck. Architecture design of a risc processor for prolog. In *Euromicro 89*, 1989.

[EC83] P. R. Eggert and K. P. Chow. Logic Programming, Graphics and Infinite Terms. Technical report, Department of Computer Science, University of California at Santa Barbara, 1983.

[Erb95] G. Erbach. ProFIT: Prolog with Features, Inheritance and Templates. In *Proceedings of the 7th Conference of the European Chapter of the Association for Computational Linguistics*, pages 180–187, Dublin, Ireland, 1995.

[GC84] F. Giannesini and J. Cohen. Parser Generation and Grammar Manipulation Using prolog’s Infinite Trees. *Journal of Logic Programming*, 3:253–265, 1984.

[Hol92] C. Holzbaur. Metastructures vs. Attributed Variables in the Context of Extensible Unification. In *1992 International Symposium on Programming Language Implementation and Logic Programming*, pages 260–268. LNCS631, Springer Verlag, August 1992.

[JS86] J. Jaffar and Peter J. Stuckey. Semantics of Infinite Tree Logic Programming. *Theoretical Computer Science*, 46:141–158, 1986.

[MM76] A. Martelli and U. Montanari. Unification in Linear Time and Space. Nota Interna B76-16, Instituto di Elaborazione della Informazione, Pisa, Italy, 1976.

[MM82] A. Martelli and U. Montanari. An efficient unification algorithm. *ACM Transactions on Programming Languages and Systems*, 4(2):258–282, 1982.

[PRO] The PROLOG IV Team. *PROLOG IV Manual*.

[PS84] C. Pollard and I. A. Sag. *Head-Driven Phrase Structure Grammar*. University of Chicago Press, Chicago, 1984.

[PW78] M.S. Paterson and M.N. Wegman. Linear unification. *Journal of Computer and System Sciences*, 17:348–375, 1978.